Click the "!!!" button to execute the commands on this worksheet.

Math 2130 Fall 2019

Quiz #3 Answer Key - In Maple!

(9/16/2019)

8am Section 101

```
> # Clear memory and load up tools...

  restart;
  with(VectorCalculus):
  with(plots):
```

#1) Let r(t) be a vector valued function (defined below) with domain $-1 \leq t \leq 7$.

```
> r := t -> <t^3+1,sin(t),5*t-3>:
  'r(t)' = r(t);
```

$$r(t) = (t^3+1)e_x + (\sin(t))e_y + (5t-3)e_z \quad \textbf{(1)}$$

#1a) Find a parameterization for the line tangent to r(t) at t=0.

We need to find the derivative of r(t) and then we need to plug in t=0.

Then the tangent line is parameterized by l(t) = r(0) + r'(0) t.

```
> diff(r(t),t);
```

$$(3t^2)e_x + (\cos(t))e_y + (5)e_z \quad \textbf{(2)}$$

```
> r(0);
```

$$(1)e_x + (0)e_y + (-3)e_z \quad \textbf{(3)}$$

```
> simplify(subs(t=0,diff(r(t),t)));
```

$$(0)e_x + (1)e_y + (5)e_z \quad \textbf{(4)}$$

```
> tline := t -> r(0)+simplify(subs(u=0,diff(r(u),u)))*t:
  'tline(t)' = tline(t);
```

$$tline(t) = (1)e_x + (t)e_y + (5t-3)e_z \quad \textbf{(5)}$$

We can graph this tangent line along with r(t). I'll graph the tangent in red and give it a domain of $-2 \leq t \leq 2$ so we can see a little bit of the tangent "before" and "after" the point it's tangent to (i.e., tline(0)=r(0)).

```
> A := spacecurve(r(t),t=-1..7):
  B := spacecurve(tline(t),t=-2..2,color=red):
  display(A,B);
```



#1b) Find the unit tangent.

We can either do this directly (remembering the "Norm" computes lengths and that T(t) = r'(t) / |r'(t)|) or we can use the VectorCalculus procedure named TNBFrame. TNBFrame computes a list containing T(t), N(t), B(t). We can use "[1]" to pick off the first element of that list (i.e., T(t)).

```
> simplify(diff(r(t),t)/Norm(diff(r(t),t)));
```

$$\left(\frac{3\,t^2}{\sqrt{25+9\,t^4+\cos(t)^2}}\right)e_x+\left(\frac{\cos(t)}{\sqrt{25+9\,t^4+\cos(t)^2}}\right)e_y+\left(\frac{5}{\sqrt{25+9\,t^4+\cos(t)^2}}\right)e_z \quad (6)$$

...or...

```
> simplify(TNBFrame(r(t),t)[1]);
```

(7)

$$\left[\begin{array}{c}\frac{3\,t^2}{\sqrt{25+9\,t^4+\cos(t)^2}}\\ \frac{\cos(t)}{\sqrt{25+9\,t^4+\cos(t)^2}}\\ \frac{5}{\sqrt{25+9\,t^4+\cos(t)^2}}\end{array}\right] \tag{7}$$

#1c) Set up an integral that computes the arc length of our curve.

Again, we can either do this directly using $\int_a^b |r'(t)|\,\mathrm{d}t$ or we can use the built in function VectorCalculus function ArcLength.

In either case, Maple can't find an exact answer (in terms of known functions).
If we want, we can apply evalf to get a decimal approximation.

```
> int(Norm(diff(r(t),t)),t=-1..7);
```

$$\int_{-1}^{7} \sqrt{25+9\,t^4+\cos(t)^2}\,\mathrm{d}t \tag{8}$$

...or...

```
> ArcLength(r(t),t=-1..7);
```

$$\int_{-1}^{7} \sqrt{25+9\,t^4+\cos(t)^2}\,\mathrm{d}t \tag{9}$$

```
> evalf(ArcLength(r(t),t=-1..7));
```

$$355.7363983 \tag{10}$$

#2) Our velocity function and initial position are given below:

```
> v := t -> <3*t^2,-2*t,4>:
  'v(t)' = v(t);

  r0 := <3,2,1>;
```

$$v(t) = \left(3\,t^2\right)e_x + (-2\,t)e_y + (4)e_z$$
$$r0 := (3)e_x + (2)e_y + (1)e_z \tag{11}$$

#2a) Find acceleration.

This is just the derivative of the velocity function.

```
> diff(v(t),t);
```

$$(6\,t)e_x + (-2)e_y + (0)e_z \tag{12}$$

#2b) Find the position function.

We need to integrate velocity and then use the initial position to determine our arbitrary constant.

```
> r := int(v(t),t)+C;
```

$$r := C + (t^3)e_x + (-t^2)e_y + (4\,t)e_z \quad \textbf{(13)}$$

We need r at t=0 to match r0...

I can't use r(0) here since this r was defined as an expression (not a function -- there's no "->" maps to). Thus I need to use subs to substitute in t=0.

```
> subs(t=0,r)=r0;
```

$$C + (0)e_x + (0)e_y + (0)e_z = (3)e_x + (2)e_y + (1)e_z \quad \textbf{(14)}$$

```
> C := <3,2,1>;
```

$$C := (3)e_x + (2)e_y + (1)e_z \quad \textbf{(15)}$$

Our position function:

```
> 'r(t)' = r;
```

$$r(t) = (t^3 + 3)e_x + (-t^2 + 2)e_y + (1 + 4\,t)e_z \quad \textbf{(16)}$$

By the way, initial speed would be |v(0)|. This is...

```
> Norm(v(0));
```

$$4 \quad \textbf{(17)}$$

Math 2130 Fall 2019

Quiz #3 Answer Key - In Maple!

(9/16/2019)

9am Section 102

```
> # Clear memory and load up tools...

  restart;
  with(VectorCalculus):
  with(plots):
```

#1) Let r(t) be a vector valued function (defined below) with domain $-2 \le t \le 10$.

```
> r := t -> <t^3+5,3*t-1,t^2>:
  'r(t)' = r(t);
```

$$r(t) = (t^3 + 5)e_x + (3\,t - 1)e_y + (t^2)e_z \quad \textbf{(18)}$$

#1a) Find a parameterization for the line tangent to r(t) at t=2.

We need to find the derivative of r(t) and then we need to plug in t=2.

Then the tangent line is parameterized by l(t) = r(2) + r'(2) t.

```
> diff(r(t),t);
```

$$(3\,t^2)e_x + (3)e_y + (2\,t)e_z \tag{19}$$

```
> r(2);
```

$$(13)e_x + (5)e_y + (4)e_z \tag{20}$$

```
> simplify(subs(t=2,diff(r(t),t)));
```

$$(12)e_x + (3)e_y + (4)e_z \tag{21}$$

```
> tline := t -> r(2)+simplify(subs(u=2,diff(r(u),u)))*t:
  'tline(t)' = tline(t);
```

$$tline(t) = (13 + 12\,t)e_x + (5 + 3\,t)e_y + (4 + 4\,t)e_z \tag{22}$$

We can graph this tangent line along with r(t). I'll graph the tangent in red and give it a domain of $-2 \le t \le 2$ so we can see a little bit of the tangent "before" and "after" the point it's tangent to (i.e., tline(0)=r(2)).

```
> A := spacecurve(r(t),t=-2..10):
  B := spacecurve(tline(t),t=-2..2,color=red):
  display(A,B);
```



#1b) Find the unit tangent.

We can either do this directly (remembering the "Norm" computes lengths and that T(t) = r'(t) / |r'(t)|) or we can use the VectorCalculus procedure named TNBFrame. TNBFrame computes a list containing T(t), N(t), B(t). We can use "[1]" to pick off the first element of that list (i.e., T(t)).

```
> simplify(diff(r(t),t)/Norm(diff(r(t),t)));
```

$$\left(\frac{3\,t^2}{\sqrt{9\,t^4+4\,t^2+9}}\right)e_x+\left(\frac{3}{\sqrt{9\,t^4+4\,t^2+9}}\right)e_y+\left(\frac{2\,t}{\sqrt{9\,t^4+4\,t^2+9}}\right)e_z \quad (23)$$

...or...

```
> simplify(TNBFrame(r(t),t)[1]);
```

$$\left[\begin{array}{c}\frac{3\,t^2}{\sqrt{9\,t^4+4\,t^2+9}}\\ \frac{3}{\sqrt{9\,t^4+4\,t^2+9}}\\ \frac{2\,t}{\sqrt{9\,t^4+4\,t^2+9}}\end{array}\right] \quad (24)$$

#1c) Set up an integral that computes the arc length of our curve.

Again, we can either do this directly using $\int_a^b |r'(t)|\,\mathrm{d}t$ or we can use the built in function VectorCalculus function ArcLength.

Maple can find an exact answer (in terms of functions unfamiliar to us).
If we want, we can apply evalf to get a decimal approximation.

Note: I will use an inert function "Int" instead of "int" since this will cause Maple to print out the definite integral, but not actually compute the answer.

```
> Int(Norm(diff(r(t),t)),t=-2..10);
```

$$\int_{-2}^{10}\sqrt{9\,t^4+4\,t^2+9}\,\mathrm{d}t \quad (25)$$

...or...

```
> ArcLength(r(t),t=-2..10);
```

$$\frac{9130\sqrt{90409}}{2727}+\frac{1274}{135}+\frac{44\,\mathrm{EllipticK}\left(\frac{\sqrt{14}}{6}\right)}{9}-\frac{11\,\mathrm{EllipticF}\left(\frac{4}{5},\frac{\sqrt{14}}{6}\right)}{9}-\frac{16\,\mathrm{EllipticE}\left(\frac{\sqrt{14}}{6}\right)}{9}+\frac{4\,\mathrm{EllipticE}\left(\frac{4}{5},\frac{\sqrt{14}}{6}\right)}{9}-\frac{11\,\mathrm{EllipticF}\left(\frac{20}{101},\frac{\sqrt{14}}{6}\right)}{9} \quad (26)$$

$$+\frac{4\,\mathrm{EllipticE}\left(\frac{20}{101}, \frac{\sqrt{14}}{6}\right)}{9}$$

```
> evalf(ArcLength(r(t),t=-2..10));
```

$$1021.314877 \quad (27)$$

#2) Our velocity function and initial position are given below:

```
> v := t -> <2*t,3*t^2-1,exp(t)>:
  'v(t)' = v(t);

  r0 := <2,0,-5>;
```

$$v(t) = (2\,t)e_x + (3\,t^2 - 1)e_y + (e^t)e_z$$

$$r0 := (2)e_x + (0)e_y + (-5)e_z \quad (28)$$

#2a) Find acceleration.

This is just the derivative of the velocity function.

```
> diff(v(t),t);
```

$$(2)e_x + (6\,t)e_y + (e^t)e_z \quad (29)$$

#2b) Find the position function.

We need to integrate velocity and then use the initial position to determine our arbitrary constant.

```
> r := int(v(t),t)+C;
```

$$r := C + (t^2)e_x + (t^3 - t)e_y + (e^t)e_z \quad (30)$$

We need r at t=0 to match r0...

I can't use r(0) here since this r was defined as an expression (not a function -- there's no "->" maps to). Thus I need to use subs to substitute in t=0.

```
> subs(t=0,r)=r0;
```

$$C + (0)e_x + (0)e_y + (e^0)e_z = (2)e_x + (0)e_y + (-5)e_z \quad (31)$$

```
> C := <2,0,-5>-<0,0,1>;
```

$$C := (2)e_x + (0)e_y + (-6)e_z \quad (32)$$

Our position function:

```
> 'r(t)' = r;
```

$$r(t) = (t^2 + 2)e_x + (t^3 - t)e_y + (-6 + e^t)e_z \quad (33)$$

By the way, initial speed would be |v(0)|. This is...

```
> Norm(v(0));
```

$$\sqrt{2} \tag{34}$$